

14 Vector Calculus

14.1 Parametric Equations of Curves

In this section we will see how Maple V can be used to plot interesting curves using parametric representation.

First of all consider the circle centered at the origin with radius 2. The equation for this curve is

$$x^2 + y^2 = 4$$

If you wish plot this curve then you soon become aware that the set of points represented by the above equation cannot satisfy a relationship of the form $y = f(x)$ where f is a single valued function (the vertical line test fails). The graph of the function

$$f_1(x) = \sqrt{4 - x^2}$$

is the upper semi-circle of radius 2.

```
> P1 := plot(sqrt(4-x^2), x=-2..2):";
```

Whereas the following plot gives the bottom half:

```
> P2 := plot(-sqrt(4-x^2), x=-2..2):";
```

We can now use **display** to put both plots together. See Figure 70.

```
> with(plots):
```

```
> display({P1,P2}, scaling =constrained);
```

We can obtain a parameterization for the entire circle. Since the following identity is true for all real values of t




Figure 70: Circle of Radius 2




Figure 71: Parabola $x = t$, $y = t^3 - 2t$

$$\sin(t)^2 + \cos(t)^2 = 1$$

it follows that $x = 2 \sin(t)$, $y = 2 \cos(t)$ satisfy the equation

$$x^2 + y^2 = 4$$

for all values of t . Moreover, the image of the mapping defined by the parametric equations $x = 2 \cos(t)$, $y = 2 \sin(t)$ generate the entire circle, which is shown in Figure 70, as t varies for 0 to 2π with one plot statement using the **parametric** plot syntax.

```
> plot([2*cos(t), 2*sin(t), t=0..2*Pi], scaling=constrained);
```

If a curve is given as an explicit function of x or y then it is easy to write a parameterization: if a curve is described by the relation $y = f(x)$, for $a \leq x \leq b$, then a parameterization for this curve is $x = t$, $y = f(t)$, for $a \leq t \leq b$, and, similarly, if a curve is described as $x = g(y)$, then the parameterization can be of the form $x = g(t)$, $y = t$. For example, suppose

$$y = x^3 - 2x$$

is given. Then either of the following plot commands will give the Figure 71. for any range, say, $-3 \leq x \leq 3$.

```
> plot(x^3-2*x,x=-3..3);
> plot([t,t^3-2*t,t=-3..3]);
```

Similarly if $x = y^2 - y$ is the function you wish to plot then the following command does the trick. See Figure 72.

```
> plot([t^2-t,t,t=-2..2]);
```




Figure 72: Parabola $x = t^2 - t$, $y = t$




Figure 73: Line $x = 1 + 2t$, $y = 3 + 5t$

Straight lines are very easy to parameterize. For example if $y = mx + b$ is the equation of the line then $x = t$, $y = mt + b$ gives a parameterization. But you can sometimes write the parametric equations directly from their geometric description. For example, a portion of the straight line through the point $(1, 3)$ parallel to the vector $\langle 2, 5 \rangle$ has parametric equations $x = 1 + 2t$, $y = 3 + 5t$, is shown in Figure 73 and can be plotted by the command:

```
> plot([1+2*t,3+5*t,t=-1..1]);
```

On occasion it is possible to plot curves that would be very complicated to express in even implicit algebraic terms. For example, the following curve is known as a Lissajous Figure and is shown in Figure 74.



```
> plot([cos(3*t),sin(5*t),t=0..2*Pi],scaling=constrained);
```

Solutions of differential equations are usually represented in parametric form. For example $x = \cos(2t)$, $y = -2\sin(2t)$ is a solution of the initial value problem:

$$\begin{aligned} \frac{d}{dt}x(t) &= -x(t) \\ \frac{d}{dt}y(t) &= -4x(t) \\ x(0) &= 1, \quad y(0) = 0 \end{aligned}$$

as can easily be verified. A plot of this curve is given by

```
> plot([cos(2*t),-2*sin(2*t),t=0..Pi],scaling = constrained);
```

Figure 74: Lissajous Figure $x = \cos(3t)$, $y = \sin(5t)$ Figure 75: Ellipse $x = \cos(2t)$, $y = -2\sin(2t)$

See Figure 75. When plotting curves in three dimensions using Maple V we need to use the command **spacecurve** which is part of the **plots** package. Consider the curve written parametrically as

$$x = \cos t, \quad y = \sin t, \quad z = t.$$

It's plot, Figure 76 over the range $[0, 2\pi]$ is given by

```
> spacecurve([cos(t), sin(t), t], t=0..2*Pi);
```



Figure 76: Cylindrical Helix: $x = \cos(t)$, $y = \sin(t)$, $z = t$ 

Figure 77: A Closed Curve in Three Space

The complicated curve shown in Figure 77, is actually a simple closed curve, and can be obtained as follows:

```
> spacecurve([(4+sin(3*t))*cos(2*t), (4+sin(3*t))*sin(2*t), cos(3*t)],  
> t=0..2*Pi);
```

Exercises 14.1

1. Make a Maple V plot of the conical helix given by the parametric equations

$$x = t \cos(3t), \quad y = t \sin(3t), \quad z = t, \quad -\infty < t < \infty.$$

2. Write parametric equations for first octant portion of the curve of intersection of the sphere and cylinder

$$x^2 + y^2 + z^2 = 64, \quad x^2 + (y - 4)^2 = 16.$$

Make a Maple V plot of this curve.

3. Show that $x = \sin(2t) \cos(t)$, $y = \sin(2t) \sin(t)$, $0 \leq t \leq 2\pi$ is a parametric representation for the curve given implicitly by the equation

$$(x^2 + y^2)^3 = 4x^2y^2.$$

Plot the graph of this curve using Maple V using the parametric representation and also using the explicit representation. Is it worthwhile to obtain the parametric representation?

14.2 Parametric Equations of Surfaces

As with the implicit representation of a circle, the implicit representation of a sphere may require a piece meal approach when making plots. In order to plot the graph of the sphere

$$x^2 + y^2 + z^2 = 4$$

we could solve for z and obtain a parameterization for the lower half with

$$z = f_1(x, y) = -\sqrt{4 - x^2 - y^2}, \quad 0 \leq x^2 + y^2 \leq 4$$

and obtain a plot with **plot3d**.

```
> P1 := plot3d(-sqrt(4-x^2-y^2), x=-2..2, y=-sqrt(4-x^2)..sqrt(4-x^2),
> style=patchnogrid, numpoints=2500):";
```

In order to get the top half we type in the following:

```
> P2 := plot3d(sqrt(4-x^2-y^2), x=-2..2, y=-sqrt(4-x^2)..sqrt(4-x^2),
> style=patchnogrid, numpoints=2500):";
```

Finally, to put both plots, as shown in Figure 78, at the same time we write.

```
> with(plots):
> display({P1, P2}, axes=normal, scaling=constrained);

> with(plots):
> display({P1, P2}, scaling =constrained);
```




Figure 78: Sphere of Radius 2




Figure 79: Cone Plot with **implicitplot3d**

This is analogous to the procedure that was use in plotting a circle when its implicit equation is given in rectangular coordinates. Just as with the situation with a circle, there is a parameterization which enables us to plot the graph with just one set of defining equations. From a well known trigonometric identity you can see that

$$x = 2 \sin(u) \cos(v), \quad y = 2 \sin(u) \sin(v), \quad z = 2 \cos(u)$$

provides a parametric representation for the entire sphere if $0 \leq u \leq \pi$, $0 \leq v \leq 2\pi$. The sphere can thus be plotted, as is also shown in Figure 79, as follows:

```
> plot3d([2*sin(u)*cos(v), 2*sin(u)*sin(v), 2*cos(u)], u=0..Pi, v=Pi..2*Pi,
> scaling=constrained, axes=normal, style=patch);
```

Example: Find a parameterizations for the bottom half of the cone that intersects the $z = 0$ plane in the circle

$$x^2 + y^2 = 4$$

and whose vertex is located a height 8 above the plane $z=0$.

Solution: An implicit equation for the entire cone is seen to be

$$4 \left(1 - \frac{1}{8} z \right)^2 = x^2 + y^2$$

and thus the following use of **implicitplot3d** produces the plot, shown in Figure 79, of the surface.

```
> implicitplot3d(4*(1-z/8)^2=x^2+y^2,x=-3..3,y=-3..3,z=-1..17);
```

You can verify that if

$$x = \left(1 - \frac{1}{8} u \right) \cos(v), y = \left(1 - \frac{1}{8} u \right) \sin(v), z = u$$

then the point (x, y, z) lies on the cone. Thus the following command plots the bottom half of the cone.

```
> plot3d([(1-u/8)*cos(v),(1-u/8)*sin(v),u],u=0..8,v=0..2*Pi);
```




Figure 80: Cone Using Parametric Equations




Figure 81: Plot of Torus

Example: Plot the torus which is created by revolving the circle

$$(x - 1)^2 + z^2 = 4$$

about the z -axis. After verifying that the following gives a parameterization for this torus

$$x = 2 \cos(u) + \cos(v) \cos(u), y = 2 \sin(u) + \cos(v) \sin(u), z = \sin(v)$$

it is easy to see that the the next command plots it. See Figure 81.

```
> plot3d([2*cos(u)+cos(v)*cos(u),2*sin(u)+cos(v)*sin(u),sin(v)],
> u=0..2*Pi,v=0..2*Pi);
```

Exercises 14.2

1. Find an implicit equation in rectangular coordinates for the surface given by

$$x = 2u \cos v, \quad y = 3u \sin v, \quad z = u.$$

Make a Maple V plot of the surface using the given parametric representation.

2. Write parametric equations for the hyperboloid of one sheet

$$x^2 + y^2 = z^2 + 1$$

and make a Maple V plot that illustrates the qualitative features of this surface.

14.3 Velocity and Acceleration Vectors in Space

Consider a curve parameterized by $t \in [a, b]$:

$$x = f(t), \quad y = g(t), \quad z = h(t).$$

```
> with(linalg):
```

```
Warning: new definition for norm
Warning: new definition for trace
```

The position vector R for this function is given by

```
> R := vector([f(t), g(t), h(t)]);
```

$$R := [f(t) \ g(t) \ h(t)]$$

The velocity vector V is obtained by differentiation, which since R is a vector requires the **map** command.

```
> V := map(diff, R, t);
```

$$V := \left[\frac{\partial}{\partial t} f(t) \quad \frac{\partial}{\partial t} g(t) \quad \frac{\partial}{\partial t} h(t) \right]$$

The magnitude of the the velocity vector is the speed $\frac{ds}{dt}$.

```
> dsdt := sqrt(dotprod(V, V));
```

$$dsdt := \left(\left(\frac{\partial}{\partial t} f(t) \right)^2 + \left(\frac{\partial}{\partial t} g(t) \right)^2 + \left(\frac{\partial}{\partial t} h(t) \right)^2 \right)^{1/2}$$

The arc length of the curve is found through integrating the speed which gives the formula:

```
> Length := Int(s, t=a..b);
```

$$Length := \int_a^b \left(\left(\frac{\partial}{\partial t} f(t) \right)^2 + \left(\frac{\partial}{\partial t} g(t) \right)^2 + \left(\frac{\partial}{\partial t} h(t) \right)^2 \right)^{1/2} dt$$

As an example consider the curve parameterized by

$$x = (4 + \sin t) \cos 3t, \quad y = (4 + \sin t), \quad z = \sin 3t, \cos(t)$$

The position vector R is

```
> R := vector([(4+sin(t))*cos(3*t), (4+sin(t))*sin(3*t), cos(t)]);
```

$$R := [(4 + \sin(t)) \cos(3t) \ (4 + \sin(t)) \sin(3t) \ \cos(t)]$$

A plot of this curve from $t = 0$ to $t = \pi$ is shown in Figure 82 and is produced by

```
> with(plots):
```

```
> spacecurve(R, t=0..2*Pi);
```

Thus the tangent vector is:

```
> V := map(diff, R, t);
```


$$V := [\cos(t)\cos(3t) - 3(4 + \sin(t))\sin(3t) \\ \cos(t)\sin(3t) + 3(4 + \sin(t))\cos(3t) - \sin(t)]$$

The square of the speed is

```
> dsdt2 := dotprod(V,V);
```

$$dsdt2 := (\cos(t)\cos(3t) - 12\sin(3t) - 3\sin(3t)\sin(t))^2 \\ + (\cos(t)\sin(3t) + 12\cos(3t) + 3\cos(3t)\sin(t))^2 + \sin(t)^2$$

```
> dsdt2 := simplify(dsdt2);
```

$$dsdt2 := -9\cos(t)^2 + 72\sin(t) + 154$$

The speed is

```
> dsdt := sqrt(dsdt2);
```

$$dsdt := \sqrt{-9\cos(t)^2 + 72\sin(t) + 154}$$

The length of the arc from $t = 0$ to $t = 2\pi$ is

```
> Length := Int(s,t=0..2*Pi);
```

$$Length := \int_0^{2\pi} \sqrt{-9\cos(t)^2 + 72\sin(t) + 154} dt$$

A ten digit approximation for this is given by

```
> Length := evalf(Length);
```

$$Length := 75.66806064$$




Figure 82: Curve in Space




Figure 83: A Non-linear Helix

We now return our expressions R , V , and $\frac{ds}{dt}$ to the general case to discuss acceleration.

```
> R := vector([f(t),g(t),h(t)]); V:= map(diff,R,t); dsdt := sqrt(dotprod(V,V));
```

$$R := [f(t) g(t) h(t)]$$

$$V := \left[\frac{\partial}{\partial t} f(t) \frac{\partial}{\partial t} g(t) \frac{\partial}{\partial t} h(t) \right]$$

$$dsdt := \sqrt{\left(\frac{\partial}{\partial t} f(t) \right)^2 + \left(\frac{\partial}{\partial t} g(t) \right)^2 + \left(\frac{\partial}{\partial t} h(t) \right)^2}$$

The unit tangent vector T is the normalized tangent vector:

```
> T := scalarmul(V, 1/dsdt);
```

$$T := \left[\frac{\frac{\partial}{\partial t} f(t)}{\sqrt{\%1}} \frac{\frac{\partial}{\partial t} g(t)}{\sqrt{\%1}} \frac{\frac{\partial}{\partial t} h(t)}{\sqrt{\%1}} \right]$$

$$\%1 := \left(\frac{\partial}{\partial t} f(t) \right)^2 + \left(\frac{\partial}{\partial t} g(t) \right)^2 + \left(\frac{\partial}{\partial t} h(t) \right)^2$$

One important fact about T and its derivative is that they are orthogonal. To see this calculate the derivative.

```
> dTdt := map(diff, T, t);
```

$$dTdt := \left[-\frac{1}{2} \frac{\left(\frac{\partial}{\partial t} f(t) \right) \%5}{\%2^{3/2}} + \frac{\%4}{\sqrt{\%2}} - \frac{1}{2} \frac{\left(\frac{\partial}{\partial t} g(t) \right) \%5}{\%2^{3/2}} + \frac{\%3}{\sqrt{\%2}} \right. \\ \left. - \frac{1}{2} \frac{\left(\frac{\partial}{\partial t} h(t) \right) \%5}{\%2^{3/2}} + \frac{\%1}{\sqrt{\%2}} \right]$$

$$\%1 := \frac{\partial^2}{\partial t^2} h(t)$$

$$\%2 := \left(\frac{\partial}{\partial t} f(t) \right)^2 + \left(\frac{\partial}{\partial t} g(t) \right)^2 + \left(\frac{\partial}{\partial t} h(t) \right)^2$$

$$\%3 := \frac{\partial^2}{\partial t^2} g(t)$$

$$\%4 := \frac{\partial^2}{\partial t^2} f(t)$$

$$\%5 := 2 \left(\frac{\partial}{\partial t} f(t) \right) \%4 + 2 \left(\frac{\partial}{\partial t} g(t) \right) \%3 + 2 \left(\frac{\partial}{\partial t} h(t) \right) \%1$$

Now take the dot product of T and its derivative:

```
> simplify(dotprod(T, dTdt));
```

0

Thus T is perpendicular to its derivative.

The acceleration vector is equal to the derivative of the velocity vector:

```
> A := map(diff, V, t);
```

$$A := \left[\frac{\partial^2}{\partial t^2} f(t) \frac{\partial^2}{\partial t^2} g(t) \frac{\partial^2}{\partial t^2} h(t) \right]$$

The radius of curvature is given by

```
> r := s/sqrt(dotprod(dTdt, dTdt));
```

The curvature is the reciprocal of the radius of curvature:

```
> kappa := 1/r;
```

Example: Find the tangential and normal components of acceleration and the curvature for the helix given by the equations for $t > 0$:

$$x = 3 \cos(t^2), y = 3 \sin(t^2), z = t^2$$

Solution: The position vector is

```
> R := vector([3*cos(t^2), 3*sin(t^2), t^2]);
```

$$R := [3 \cos(t^2) \ 3 \sin(t^2) \ t^2]$$

We aren't asked to plot this curve for this problem but we will anyway. See Figure 83

```
> plots[spacecurve](R, t=0..Pi);
```

The velocity and acceleration vectors are:

```
> V := map(diff, R, t); A := map(diff, V, t);
```

$$V := [-6 \sin(t^2) \ t \ 6 \cos(t^2) \ t \ 2 \ t]$$

$$A := [-12 \cos(t^2) \ t^2 - 6 \sin(t^2) \ -12 \sin(t^2) \ t^2 + 6 \cos(t^2) \ 2 \ t]$$

The speed and unit tangent vectors are

```
> dsdt := simplify(sqrt(simplify(dotprod(V, V))), symbolic);
```

$$dsdt := 2\sqrt{10}t$$

```
> T := scalarmul(V, 1/dsdt);
```

$$T := \left[-\frac{3}{10} \sqrt{10} \sin(t^2) \ \frac{3}{10} \sqrt{10} \cos(t^2) \ \frac{1}{10} \sqrt{10} \right]$$

Using vector analysis we know that the tangential component of acceleration is

$$\text{dotprod}(A, T) \cdot T.$$

```
> Atan := scalarmul(T, dotprod(A, T));
```

$$\begin{aligned} \text{Atan} := & \left[-\frac{3}{10} \%1 \sqrt{10} \sin(t^2) \ \frac{3}{10} \%1 \sqrt{10} \cos(t^2) \ \frac{1}{10} \%1 \sqrt{10} \right] \\ & \%1 := -\frac{3}{10} (-12 \cos(t^2) t^2 - 6 \sin(t^2)) \sqrt{10} \sin(t^2) \\ & + \frac{3}{10} (-12 \sin(t^2) t^2 + 6 \cos(t^2)) \sqrt{10} \cos(t^2) + \frac{1}{5} \sqrt{10} \end{aligned}$$

```
> simplify(%1);
```

$$2\sqrt{10}$$

```
> Atan := map(z->subs(%1=2*sqrt(10),z),Atan);
```

$$Atan := [-6\sin(t^2) \ 6\cos(t^2) \ 2]$$

```
> absAtan := sqrt(simplify(dotprod(Atan,Atan)));
```

$$absAtan := 2\sqrt{10}$$

Of course the other formula for this quantity is $\frac{d^2s}{dt^2}$:

```
> diff(dsdt,t);
```

$$2\sqrt{10}$$

If you know the acceleration vector and its tangential component then by vector subtraction you can get the normal component of acceleration.

```
> Anormal := evalm(A-Atan);
```

$$Anormal := [-12\cos(t^2)t^2 - 12\sin(t^2)t^2 \ 0]$$

The magnitude of this component is thus

```
> simplify(sqrt(simplify(dotprod(Anormal,Anormal))),symbolic);
```

$$12t^2$$

This means that the curvature is calculated by

```
> kappa := "/dsdt^2;
```

$$\kappa := \frac{3}{10}$$

Another way to obtain the normal component is from the formula:

$$\frac{ds}{dt} \cdot dT/dt.$$

```
> dTdt := map(diff,T,t);
```

$$dTdt := \left[-\frac{3}{5}\sqrt{10}\cos(t^2)t - \frac{3}{5}\sqrt{10}\sin(t^2)t \ 0 \right]$$

```
> scalarmul(dTdt,dsdt);
```

$$[-12\cos(t^2)t^2 - 12\sin(t^2)t^2 \ 0]$$

This agrees with the previous expression for the normal component of acceleration.

Exercises 14.3

1. If

$$\mathbf{R}(t) = \sin(2t)\mathbf{i} + 3\cos(3t)\mathbf{j} + t^2\mathbf{k},$$

find \mathbf{V} , \mathbf{A} , $\frac{ds}{dt}$, and the approximate value of the arc length of the curve as t varies from 0 to 2π . Plot the curve over this interval.

14.4 Vector Fields

There are several graphing procedures in Maple V that help you to study vector fields.

```
> with(plots):
```

Within the **plots** package there is the command **gradplot** that allows you to plot the gradient vector field for a function of two variables. For a plot of the gradient vector field of the scalar function $f(x, y) = x^2 + 2y^2$ see Figure 84.




Figure 84: Gradient Plot of $x^2 + 2y^2$




Figure 85: Gradient Plot with some Level Curves

```
> Q := gradplot(x^2+2*y^2,x=-3..3,y=-3..3):";
```

This command along with **implicitplot** allows you to plot the vector field along with several level curves. See Figure 85.

```
> P.1 := implicitplot(x^2+2*y^2=1,x=-3..3,y=-3..3):";
> for i from 1 to 3 do P.(2*i+1) := implicitplot(x^2+2*y^2=2*i+1,
> x=-3..3,y=-3..3):od:
> display({Q,seq(P.(2*i-1),i=1..4)});
```

If you wish to view the flow the vector field in two dimensions, then it is useful to utilize the tools that are part of the **DEtools** package.

```
> with(DEtools):
```

We first use **grad** to calculate the gradient vector field of the function.

```
> f := x^2+2*y^2;
```

$$f := x^2 + 2y^2$$

The **linalg** package needs to be made available for some of the commands that follow.

```
> with(linalg):
```

```
Warning: new definition for norm
Warning: new definition for trace
```

```
> X := grad(f,[x,y]);
```

$$X := [2x \ 4y]$$

However, to use this vector field as an argument for **DEplot2** we must convert it to the type **list**.

```
> X := convert(X,list);
```

$$X := [2x, 4y]$$

The next command plots the gradient vector field along with the three flowlines which start at the points $(-1.5, 1.1)$, $(1.1, 0.83)$, and $(0.8, 0.9)$. See Figure 86.




Figure 86: Gradient Vector Field and some Flow Lines




Figure 87: Gradient Vector Field with Level Curves and Flow Lines

```
> Flow := DEplot2(X,[x,y],-0.5..0.5,
> {[0,-1.5,1.1],[0,1.1,0.83],[0,0.8,0.9]}):";
```

Now if we want some level curves for $f(x, y)$, along with the vector field and flowlines, we do something like the following. See Figure 87.

```
> P.1 := implicitplot(x^2+2*y^2=1,x=-4..3,y=0..9):";
> for i from 1 to 5 do P.(2*i) := implicitplot(x^2+2*y^2=(2*i)^2,
> x=-4..3,y=0..9):od:
> display({Flow,seq(P.(2*i),i=1..5)},scaling=constrained);
```

It is also possible to plot two and three dimensional vector fields using **fieldplot** and **fieldplot3d**. For example to plot in Figure 88 of the vector field

$$\langle 2x, y, x - z \rangle$$

execute the following commands:

```
> X := vector([2*x,y,x-z]);
```

$$X := [2xyx - z]$$

```
> fieldplot3d(X,x=-3..3,y=-3..3,z=-3..3);
```

Exercises 14.4

1. Plot the level curves

$$f(x, y) = x^2 - xy + y^2 = j, \text{ for } j = 1, 2, 3$$

on a single Maple V plot with $-4 \leq x \leq 4$, $-4 \leq y \leq 4$. Make a plot of the gradient vector field along with flow lines through the points $(-1.5, 1)$, $(1, 1)$, $(1, 2)$, and $(1, -3)$.




Figure 88: A Three Dimensional Field Plot

2. Plot flow lines for the vector field

$$\mathbf{F}(x, y) = \langle -2y, x \rangle$$

that go through points $(1, 0)$, $(2, 0)$, and $(3, 0)$.

14.5 Line Integrals with Maple V

In this section we compute line integrals of vector fields over curves in two and three dimensional space.

```
> with(linalg):
```

For a start lets compute the line integral of the vector field $\langle x^2, x \rangle$ over the straight line that connects (0,0) to (1,0).

```
> V := vector([x^2,x]);
```

$$V := [x^2, x]$$

```
> r := vector([t,0]);
```

$$r := [t, 0]$$

```
> dr := map(diff,r,t);
```

$$dr := [1, 0]$$

```
> B := subs({x=t,y=0},dotprod(V,dr));
```

$$B := t^2$$

```
> Int(V,r) = int(B,t=0..1);
```

$$\int V dr = \frac{1}{3}$$

As another example let's integrate the vector field

$$X = \langle xy, yz, xz \rangle$$

over the twisted cubic, which is given parametrically by

$$x = t, \quad y = t^2, \quad z = t^3,$$

from $(-1, 1, -1)$ to $(1, 1, 1)$.

```
> V := vector([x*y,y*z,x*z]); r := vector([t,t^2,t^3]);
```

$$V := [xy, yz, xz]$$

$$r := [t, t^2, t^3]$$

```
> dr := map(diff,r,t);
```

$$dr := [1, 2t, 3t^2]$$

```
> B := subs({x=t,y=t^2,z=t^3},dotprod(V,dr));
```

$$B := t^3 + 5t^6$$

```
> Int(V,r) = int(B,t=-1..1);
```


$$\int V \, dr = \frac{10}{7}$$

Exercises 14.5

1. Let $\mathbf{F}(x, y, z) = \langle x^3y, xy, yz \rangle$. Find the work done by \mathbf{F} along a circle of radius 5 centered at the origin in the yz plane.
2. Evaluate the line integral of the vector field $\mathbf{F}(x, y, z) = \langle xye^{-y}, xz, \sin(z) \rangle$ along the parabola $z = x^2$, $y = 0$, between $(-1, 0, 1)$ and $(1, 1, 1)$.

14.6 Divergence and Curl in Maple V

We can compute the curl and divergence of a vector field by using commands that are contained in the **linalg** package.

```
> with(linalg):
```

```
Warning: new definition for    norm
```

```
Warning: new definition for    trace
```

As an example we will calculate the divergence and curl of the vector field

$$\langle \sin x, \exp xyz, x + \sin z \rangle.$$

```
> F := vector([sin(x),exp(x*y*z),x+sin(z)]);
```

$$F := [\sin(x) e^{(xyz)} x + \sin(z)]$$

```
> divF := diverge(F,[x,y,z]);
```

$$\text{div} F := \cos(x) + xz e^{(xyz)} + \cos(z)$$

```
> curlF := curl(F,[x,y,z]);
```

$$\text{curl} F := [-x y e^{(xyz)} - 1 y z e^{(xyz)}]$$

If we need to find the value of these quantities at a particular point, say $(0, 1, \pi)$, then we can use **subs** and **eval**.

```
> ValuedivF := subs(x=0,y=1,z=Pi,divF);
```

$$\text{Valuediv} F := \cos(0) + \cos(\pi)$$

```
> ValuedivF := simplify(ValuedivF);
```

$$\text{Valuediv} F := 0$$

```
> ValuecurlF := subs(x=0,y=1,z=Pi,eval(curlF));
```

$$\text{Valuecurl} F := [0 - 1\pi e^0]$$

```
> ValuecurlF := map(simplify,ValuecurlF);
```

$$\text{Valuecurl} F := [0 - 1\pi]$$

Note that there was a little difference in finding these values. The divergence is a scalar expression and so a simple use of **simplify** was able to simplify it. On the other hand the curl of a vector field is a vector and required the use of **eval** and **map** to simplify it completely.

Exercises 14.6

1. Compute the divergence and curl for the vector field

$$\mathbf{F}(x, y, z) = \langle x^2 \sin(yz), y^2 - xz, 2x^2 z e^{z^2} \rangle.$$

Evaluate the divergence and curl at the point $(1, 2, -1)$.

14.7 Surface Integrals with Maple V

Maple V can be useful in calculating surface integrals of vector fields over a parameterized surface. Let

$$\mathbf{F}(x, y, z) = \langle f(x, y, z), g(x, y, z), h(x, y, z) \rangle$$

denote an arbitrary vector field in three dimensional space. We will define it to a Maple V session in the following. We assume that a surface S is parameterized with

$$\mathbf{R}(u, v) = \langle x(u, v), y(u, v), z(u, v) \rangle.$$

```
> f := 'f':g:= 'g':h:='h':
> with(linalg):
Warning: new definition for    norm
Warning: new definition for    trace

> F := vector([f(x,y,z),g(x,y,z),h(x,y,z)]);
```

$$F := [f(x, y, z) \ g(x, y, z) \ h(x, y, z)]$$

The surface S is defined by:

```
> R := vector([x(u,v),y(u,v),z(u,v)]);
```

$$R := [x(u, v) \ y(u, v) \ z(u, v)]$$

The tangent vectors to the level curves $v = \text{constant}$ and $u = \text{constant}$ are computed as follows:

```
> Ru := map(diff,R,u);
```

$$Ru := \left[\frac{\partial}{\partial u} x(u, v) \ \frac{\partial}{\partial u} y(u, v) \ \frac{\partial}{\partial u} z(u, v) \right]$$

```
> Rv := map(diff,R,v);
```

$$Rv := \left[\frac{\partial}{\partial v} x(u, v) \ \frac{\partial}{\partial v} y(u, v) \ \frac{\partial}{\partial v} z(u, v) \right]$$

The normal, **NS**, to the surface for the given parameterization is defined by the following.

```
> NS := crossprod(Ru,Rv);
```

$$NS := \left[\left(\frac{\partial}{\partial u} y(u, v) \right) \left(\frac{\partial}{\partial v} z(u, v) \right) - \left(\frac{\partial}{\partial u} z(u, v) \right) \left(\frac{\partial}{\partial v} y(u, v) \right) \right. \\ \left(\frac{\partial}{\partial u} z(u, v) \right) \left(\frac{\partial}{\partial v} x(u, v) \right) - \left(\frac{\partial}{\partial u} x(u, v) \right) \left(\frac{\partial}{\partial v} z(u, v) \right) \\ \left. \left(\frac{\partial}{\partial u} x(u, v) \right) \left(\frac{\partial}{\partial v} y(u, v) \right) - \left(\frac{\partial}{\partial u} y(u, v) \right) \left(\frac{\partial}{\partial v} x(u, v) \right) \right]$$

The flux integral

$$\int_S \mathbf{F} \cdot \mathbf{NS} \, dA$$

for the vector field can now be computed

```
> Int(F,R=S..NULL) = Int(Int(subs(x=x(u,v),y=y(u,v),z=z(u,v),
> dotprod(F,NS)),u=a..b),v=c..d);
```

$$\begin{aligned}
\int_S F dR = & \int_c^d \int_a^b f(x(u, v), y(u, v), z(u, v)) \left(\left(\frac{\partial}{\partial u} y(u, v)(u, v) \right) \left(\frac{\partial}{\partial v} z(u, v)(u, v) \right) \right. \\
& - \left(\frac{\partial}{\partial u} z(u, v)(u, v) \right) \left(\frac{\partial}{\partial v} y(u, v)(u, v) \right) \Bigg) + g(x(u, v), y(u, v), z(u, v)) \left(\right. \\
& \left(\frac{\partial}{\partial u} z(u, v)(u, v) \right) \left(\frac{\partial}{\partial v} x(u, v)(u, v) \right) \\
& - \left(\frac{\partial}{\partial u} x(u, v)(u, v) \right) \left(\frac{\partial}{\partial v} z(u, v)(u, v) \right) \Bigg) + h(x(u, v), y(u, v), z(u, v)) \left(\right. \\
& \left(\frac{\partial}{\partial u} x(u, v)(u, v) \right) \left(\frac{\partial}{\partial v} y(u, v)(u, v) \right) \\
& - \left(\frac{\partial}{\partial u} y(u, v)(u, v) \right) \left(\frac{\partial}{\partial v} x(u, v)(u, v) \right) \Bigg) dudv
\end{aligned}$$

Example: Evaluate the surface integral of the vector field

$$F = \langle x, y, -2z \rangle$$

over the upper half of the sphere

$$x^2 + y^2 + z^2 = a^2$$

Solution: One parameterization is given by

$$x = a \cos(u) \sin(v), y = a \sin(u) \sin(v), z = a \sin(v)$$

```
> with(linalg):
```

```
Warning: new definition for    norm Warning: new
definition for    trace
```

```
> F := vector([x,y,-2*z]);
```

$$F := [xy - 2z]$$

```
> R := vector([a*cos(u)*cos(v), a*sin(u)*cos(v), a*sin(v)]);
```

$$R := [a \cos(u) \cos(v) \ a \sin(u) \cos(v) \ a \sin(v)]$$

```
> Ru := map(diff,R,u);
```

$$Ru := [-a \sin(u) \cos(v) \ a \cos(u) \cos(v) \ 0]$$

```
> Rv := map(diff,R,v);
```

$$Rv := [-a \cos(u) \sin(v) \ -a \sin(u) \sin(v) \ a \cos(v)]$$

```
> NS := crossprod(Ru,Rv);
```

$$NS := [a^2 \cos(u) \cos(v)^2 a^2 \sin(u) \cos(v)^2 \\ a^2 \sin(u)^2 \cos(v) \sin(v) + a^2 \cos(u)^2 \cos(v) \sin(v)]$$

```
> NS := map(simplify, NS);
```

$$NS := [a^2 \cos(u) \cos(v)^2 a^2 \sin(u) \cos(v)^2 a^2 \cos(v) \sin(v)]$$

The integrand for the integral is

```
> H := subs(x=R[1], y=R[2], z=R[3], dotprod(NS, F));
```

$$H := a^3 \cos(u)^2 \cos(v)^3 + a^3 \sin(u)^2 \cos(v)^3 - 2 a^3 \cos(v) \sin(v)^2$$

```
> H := simplify(H);
```

$$H := 3 a^3 \cos(v)^3 - 2 a^3 \cos(v)$$

```
> int(int(H, u=0..Pi/2), v=0..2*Pi);
```

0

Thus the value of this surface integral is 0.

As another example consider the surface S which is bounded below by

$$z = x^2 + y^2$$

and above by the paraboloid

$$z = 4 - x^2 - y^2.$$

```
> with(plots):
```

We define our surface with the expressions

```
> z:= 'z': y:= 'y': x:= 'x': bottom:= x^2+y^2;
```

$$bottom := x^2 + y^2$$

```
> top:= 4 -x^2-y^2;
```

$$top := 4 - x^2 - y^2$$

and plot the surface as shown in Figure 89

```
> B:= cylinderplot(sqrt(z), theta=0..2*Pi, z=0..2):";
```

```
> T:= cylinderplot(sqrt(4-z), theta=0..2*Pi, z=2..4):";
```

```
> display3d({B,T}, axes=NORMAL, style=PATCH, scaling=CONSTRAINED,
```

```
> orientation=[45,75]);
```




Figure 89: Figure Formed by Two Paraboloids

Let us assume that the density of the material that the surface is composed of is

$$d(x, y, z) = \frac{z}{\pi} + \frac{1}{\pi}$$

and our problem is to calculate the total mass of the surface. This amounts to computing the surface integral of $d(x, y, z)$ over each of the two parts: the top and the bottom and adding the results.

```
> d := z/Pi+1/Pi;
```

$$d := \frac{z}{\pi} + \frac{1}{\pi}$$

Parameterizations for these surfaces are

```
> Rbottom := vector([u,v,u^2+v^2]);
```

$$R_{bottom} := [u \ v \ u^2 + v^2]$$

```
> Rtop := vector([u,v,4-u^2-v^2]);
```

$$R_{top} := [u \ v \ 4 - u^2 - v^2]$$

The normal for the bottom will now be computed.

```
> Rbottomu := map(diff,Rbottom,u); Rbottomv:= map(diff,Rbottom,v);
```

$$R_{bottomu} := [1 \ 0 \ 2u]$$

$$R_{bottomv} := [0 \ 1 \ 2v]$$

```
> Nbottom := crossprod(Rbottomu,Rbottomv);
```

$$N_{\text{bottom}} := [-2u - 2v \ 1]$$

For this problem below we want N_{bottom} to have z component pointing down so write:

```
> Nbottom := scalarmul(Nbottom, -1);
```

$$N_{\text{bottom}} := [2u \ 2v \ -1]$$

```
> magNBottom := sqrt(dotprod(Nbottom, Nbottom));
```

$$\text{magNBottom} := \sqrt{4u^2 + 4v^2 + 1}$$

The integrand is thus

```
> H := subs(x=u, y=v, z=u^2+v^2, d)*magNBottom;
```

$$H := \left(\frac{u^2 + v^2}{\pi} + \frac{1}{\pi} \right) \sqrt{4u^2 + 4v^2 + 1}$$

It follows that a double integral which is equal to the mass of the bottom surface is given by the following:

```
> Bmass := Int(Int(H, v=-sqrt(2-u^2)..sqrt(2-u^2)), u=-sqrt(2)..sqrt(2));
```

$$B_{\text{mass}} := \int_{-\sqrt{2}}^{\sqrt{2}} \int_{-\sqrt{2-u^2}}^{\sqrt{2-u^2}} \left(\frac{u^2 + v^2}{\pi} + \frac{1}{\pi} \right) \sqrt{4u^2 + 4v^2 + 1} \, dv \, du$$

This integral is pretty hard to evaluate even with Maple V

```
> value(Bmass);
```

$$\begin{aligned} & \int_{-\sqrt{2}}^{\sqrt{2}} \frac{3}{64} (24u^2 \sqrt{2-u^2} + 24 \ln(2\sqrt{2-u^2} + 3)u^2 + 66\sqrt{2-u^2} \\ & + 16u^4 \ln(2\sqrt{2-u^2} + 3) + 5 \ln(2\sqrt{2-u^2} + 3)) / \pi - \frac{3}{64} (-24u^2 \sqrt{2-u^2} \\ & + 24 \ln(-2\sqrt{2-u^2} + 3)u^2 - 66\sqrt{2-u^2} + 16u^4 \ln(-2\sqrt{2-u^2} + 3) \\ & + 5 \ln(-2\sqrt{2-u^2} + 3)) / \pi \, du \end{aligned}$$

Thus Maple V was unable to compute the exact value of this integral in its present form. But we can get a numerical approximation using **evalf**.

```
> evalf("");
```

9.300000000

If we use our head a little bit we can simplify the preceding double integral with polar coordinates, then ...

```
> Bmass := Int(Int((r^2+1)/Pi*sqrt(4*r^2+1)*r, r=0..sqrt(2)),
> theta=0..2*Pi);
```

$$B_{\text{mass}} := \int_0^{2\pi} \int_0^{\sqrt{2}} \frac{(r^2 + 1) \sqrt{4r^2 + 1} r}{\pi} \, dr \, d\theta$$

```
> Bmass := value("");
```

$$B_{mass} := \frac{93}{10}$$

Now we do the same thing for the top

```
> Rtopu := map(diff,Rtop,u); Rtopv := map(diff,Rtop,v);
```

$$R_{topu} := [10 - 2u]$$

$$R_{topv} := [01 - 2v]$$

```
> Ntop := crossprod(Rtopu,Rtopv);
```

$$N_{top} := [2u2v1]$$

```
> magNtop := sqrt(dotprod(Ntop,Ntop));
```

$$magN_{top} := \sqrt{4u^2 + 4v^2 + 1}$$

```
> H := subs(x=u,y=v,z=4-u^2-v^2,d)*magNtop;
```

$$H := \left(\frac{4 - u^2 - v^2}{\pi} + \frac{1}{\pi} \right) \sqrt{4u^2 + 4v^2 + 1}$$

Converting to polar coordinates we have

```
> H := subs(u=r*cos(theta),v=r*sin(theta),H);
```

$$H := \left(\frac{4 - r^2 \cos(\theta)^2 - r^2 \sin(\theta)^2}{\pi} + \frac{1}{\pi} \right) \sqrt{4r^2 \cos(\theta)^2 + 4r^2 \sin(\theta)^2 + 1}$$

```
> H := simplify(H);
```

$$H := -\frac{\sqrt{4r^2 + 1}(-5 + r^2)}{\pi}$$

```
> Tmass := Int(Int(H*r,r=0..sqrt(2)),theta=0..2*Pi);
```

$$T_{mass} := \int_0^{2\pi} \int_0^{\sqrt{2}} -\frac{\sqrt{4r^2 + 1}(-5 + r^2)r}{\pi} dr d\theta$$

```
> Tmass := value(Tmass);
```

$$T_{mass} := \frac{167}{10}$$

```
> Total := Bmass + Tmass;
```


$$Total := 26$$

We conclude that the total mass of the surface is 26.

Let us continue with the surface that was defined in the last example and assume that we have a vector field

$$\mathbf{F} = \langle x, y, 1 \rangle$$

given. Now the surface integral of this vector field over the bottom surface is

```
> F := vector([x,y,1]);
```

$$F := [x \ y \ 1]$$

We now make use of the fact that we oriented the normal vector for the bottom surface.

```
> H := subs(x=u,y=v,z=sqrt(u^2+v^2),dotprod(F,Nbottom));
```

$$H := 2u^2 + 2v^2 - 1$$

```
> Bflux :=
```

```
> Int(Int(H,v=-sqrt(2-u^2)..sqrt(2-u^2)),u=-sqrt(2)..sqrt(2));
```

$$Bflux := \int_{-\sqrt{2}}^{\sqrt{2}} \int_{-\sqrt{2-u^2}}^{\sqrt{2-u^2}} (2u^2 + 2v^2 - 1) dv du$$

```
> Bflux := value(Bflux);
```

$$Bflux := 2\pi$$

Now we do the same thing using the top.

```
> H:= subs(x=u,y=v,z=sqrt(4-u^2-v^2),dotprod(F,Ntop));
```

$$H := 2u^2 + 2v^2 + 1$$

```
> Tflux := Int(Int(H,v=-sqrt(2-u^2)..sqrt(2-u^2)),u=-sqrt(2)..sqrt(2));
```

$$Tflux := \int_{-\sqrt{2}}^{\sqrt{2}} \int_{-\sqrt{2-u^2}}^{\sqrt{2-u^2}} (2u^2 + 2v^2 + 1) dv du$$

```
> Tflux := value("");
```

$$Tflux := 6\pi$$

```
> SurfaceFlux := Tflux+Bflux;
```

$$SurfaceFlux := 8\pi$$

We conclude that the flux through the surface due to our vector field is 8π . Can we verify this value using the Divergence Theorem? We start by computing the divergence of F .

```
> divF := diverge(F,[x,y,z]);
```

$$\operatorname{div} F := 2$$

```
> SurfaceFlux := Int(Int(Int(divF, z=bottom..top), y=-sqrt(2-x^2)..
> sqrt(2-x^2)), x=-sqrt(2)..sqrt(2));
```

$$\text{SurfaceFlux} := \int_{-\sqrt{2}}^{\sqrt{2}} \int_{-\sqrt{2-x^2}}^{\sqrt{2-x^2}} \int_{x^2+y^2}^{4-x^2-y^2} 2 \, dz \, dy \, dx$$

```
> SurfaceFlux := value(SurfaceFlux);
```

$$\text{SurfaceFlux} := 8\pi$$

This verifies Gauss's Theorem for this problem since both methods for computing the surface flux yield the same answer.

Example: Verify Stoke's Theorem for

$$F = \langle 2x - y, -yz^2, -y^2z \rangle,$$

where S is the upperhalf surface of the sphere

$$x^2 + y^2 + z^2 = 1$$

and C is the boundary.

Solution: First we calculate the line integral of the vector field over C . The circle C is given by

```
> R := vector([cos(t), sin(t), 0]);
```

$$R := [\cos(t) \sin(t) 0]$$

We now prepare to find the value of the integral that occurs on the right hand side of Stoke's Theorem.

```
> dR := map(diff, R, t);
```

$$dR := [-\sin(t) \cos(t) 0]$$

```
> F := vector([2*x-y, -y*z^2, -y^2*z]);
```

$$F := [2x - y - yz^2 - y^2z]$$

```
> H := subs(x=cos(t), y=sin(t), z=0, dotprod(F, dR));
```

$$H := -(2\cos(t) - \sin(t))\sin(t)$$

```
> RightSide := Int(H, t=0..2*Pi);
```

$$\text{RightSide} := \int_0^{2\pi} -(2\cos(t) - \sin(t))\sin(t) \, dt$$

```
> RightSide := value(RightSide);
```

$$RightSide := \pi$$

Now we calculate the value of the integral on the left hand side of the equality in Stoke's Theorem. First we obtain a parametrization for the upper hemisphere:

$$x = \cos(u) \cos(v), y = \sin(u) \cos(v), z = \sin(v)$$

```
> R:= vector([cos(u)*cos(v), sin(u)*cos(v),sin(v)]);
```

$$R := [\cos(u) \cos(v) \sin(u) \cos(v) \sin(v)]$$

```
> Ru := map(diff,R,u);
```

$$Ru := [-\sin(u) \cos(v) \cos(u) \cos(v) 0]$$

```
> Rv := map(diff,R,v);
```

$$Rv := [-\cos(u) \sin(v) -\sin(u) \sin(v) \cos(v)]$$

```
> NS := map(simplify,eval(crossprod(Ru,Rv)));
```

$$NS := [\cos(u) \cos(v)^2 \sin(u) \cos(v)^2 \cos(v) \sin(v)]$$

```
> curlF := curl(F,[x,y,z]);
```

$$curlF := [001]$$

```
> H:= subs(x=cos(u)*cos(v),y=sin(u)*cos(v),z=sin(v),dotprod(curlF,NS));
```

$$H := \cos(v) \sin(v)$$

```
> LeftHand := Int(Int(H,u=0..2*Pi),v=0..Pi/2);
```

$$LeftHand := \int_0^{1/2\pi} \int_0^{2\pi} \cos(v) \sin(v) du dv$$

```
> LeftHand:= value(LeftHand);
```

$$LeftHand := \pi$$

Thus the two integrals have the same value as predicted by Stoke's Theorem.

Exercises 14.7

1. Let $\mathbf{F}(x, y, z) = \langle 2xy, 3yz, -4z^2 \rangle$, and S be the sphere $x^2 + y^2 + z^2 = 25$, oriented so that the normal points outward. Verify the Divergence Theorem in this case by evaluating the flux integral of the vector field \mathbf{F} over the surface, and then by computing the triple integral of $\text{div } \mathbf{F}$ over the interior of S .
2. Compute the line integral of the vector field $\mathbf{F}(x, y, z) = \langle x^2, z - yx, xz \rangle$ over the curve of intersection of the sphere $x^2 + y^2 + z^2 = 1$, with the plane $y = z$. Assume this curve is oriented so that its projection on the xy plane is oriented clockwise. Calculate the integral directly and then use Stoke's Theorem to check your answer.